Math 2270−3
October 30, 2009
Applying least squares linear regression to obtain power law fits

How do you test for power laws?

Suppose you have a collection of n data points

$$[[x_1, y_1], [x_2, y_2], [x_3, y_3], ..., [x_n, y_n]]$$

and you expect there may be a good power−law fit

$$y = b\, x^m$$

which approximately explains how the y_i's are related to the x_i's . You would like to find the "best possible" values for b and m to make this fit. It turns out, if you take the ln−ln data, your power law question is actually just a best−line fit question:
Taking (natural) logarithms of the proposed power law yields

$$\ln(y) = \ln(b) + m\ln(x).$$

So, if we write $Y = \ln(y)$ and $X = \ln(x)$, $B = \ln(b)$, this becomes the equation of a line in the new variables X and Y:

$$Y = mX + B$$

Thus, in order for there to be a power law for the original data, the ln−ln data should (approximately) satisfy the equation of a line, and vise verse. If we get a good line fit to the ln−ln data, then the slope m of this line is the power relating the original data, and the exponential e^B of the Y−intercept is the proportionality constant b in the original relation $y = b\, x^m$. With real data it is not too hard to see if the ln−ln data is well approximated by a line, in which case the original data is well−approximated by a power law.

Astronomical example (%5.4, #40)**:** As you may know, Isaac Newton was motivated by Kepler's (observed) Laws of planetary motion to discover the notions of velocity and acceleration, i.e. differential calculus and then integral calculus, along with the inverse square law of planetary acceleration around the sun.....from which he deduced the concepts of mass and force, and that the universal inverse square law for gravitatonal attraction was the ONLY force law depending only on distance between objects, which was consistent with Kepler's observations! Kepler's three observations were that

(1) Planets orbit the sun in ellipses, with the sun at one of the ellipse foci.
(2) A planet sweeps out equal areas from the sun, in equal time intervals, independently of where it is in its orbit.
(3) The square of the period of a planetary orbit is directly proportional to the cube of the orbit's semi−major axis.

So, for roughly circular orbits, Keplers third law translates to the statement that the period t is related to the radius r, by the equation $t = b\, r^{1.5}$, for some proportionality constant b. Let's see if that's consistent with the following data:

Planet	mean distance r from sun (in astronomical units where 1=dist to earth)	Orbital period t (in earth years)
Mercury	0.387	0.241
Earth	1.	1.
Jupiter	5.20	11.86
Uranus	19.18	84.0
Pluto	39.53	248.5

Maple implementation:

```
> with(LinearAlgebra) : #LinearAlgebra command package
  with(plots) : #plotting package
> rs := Vector([0.387, 1., 5.20, 19.18, 39.53]) : #radii
  ts := Vector([.241, 1., 11.86, 84.0, 248.5]) :  #corresponding periods
  pts := [seq([rs[i], ts[i]], i = 1 ..5)] : #points in the r−t plane

> Rs := map(ln, rs) : #logs of radii
  Ts := map(ln, ts) : #logs of periods
  Rs := map(evalf, Rs) : #compute decimal values − don't forget this step for your BMI work!
  Ts := map(evalf, Ts) :
  lnpts := [seq([Rs[i], Ts[i]], i = 1 ..5)] : #points in the lnr−lnt plane
  lnpts := ⟨Rs|Ts⟩ : # a better way!  (thanks Jim S.)
> ones := Vector([1, 1, 1, 1, 1]) : #a column for the linear regression fit matrix
> A := ⟨Rs|ones⟩;
```

$$A := \begin{bmatrix} -0.9493305860 & 1 \\ 0. & 1 \\ 1.648658626 & 1 \\ 2.953868069 & 1 \\ 3.677059877 & 1 \end{bmatrix} \quad (1)$$

```
> AT := Transpose(A) :
> (AT.A)^(-1).(AT.Ts); #least squares solution, (m, ln(b))!
    #notice, our data agrees with Kepler!
```

$$\begin{bmatrix} 1.49981641316970692 \\ 0.000486890692341201970 \end{bmatrix} \quad (2)$$

```
> b := exp(0.000486890692341201970) : #recover proportionality constant
```

```
> lnplot1 := plot(1.49982·R + .0004860, R=-1..4, T=-2..6, color = black) :
    #linear regression line for lnr−lnt data
  lnplot2 := pointplot(lnpts, color = red) : #actual ln−ln data
  display({lnplot1, lnplot2}, title = ‘ln−ln line fit‘);
```




```
> plot1 := plot(b·r^1.4998, r = 0..40, t = 0..250, color = black) : #almost Kepler power law
  plot2 := pointplot(pts, color = red) : #radius−period points for planets
  display({plot1, plot2}, title = ‘Kepler and Newton were right‘);
```




```
>
```